Creating Complementary Reports:

Giving Them More Ways to See and Use the Data

Lisa Fine

United BioSource Corporation

Introduction and Example

As programmers we're often asked to provide a series of reports that will be used in conjunction with one another.

Example: Department Store Sales Reports

- A Detail report allows the reader to better understand the source of the sales.
- A Summary report allows the recipient to see the "big picture" of how the corporation is performing.

Detail and Summary Reports

2011 Sales - Detail

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334
	West	E-Commerce	$90,033
		In-Store	$189,356
			$279,389
	North	E-Commerce	$88,024
		In-Store	$99,346
			$187,370
	South	E-Commerce	$110,009
		In-Store	$107,220
			$217,229
2 / J.B. Prog & Co.	East	E-Commerce	$120,050
		In-Store	$89,499
			$209,549
	West	E-Commerce	$110,631
		In-Store	$44,250
			$154,881
	North	E-Commerce	$97,897
		In-Store	$100,171
			$198,068
	South	E-Commerce	$102,425
		In-Store	$79,465
			$181,890
3 / XYZ Warehouse	East	E-Commerce	$159,902
		In-Store	$49,126
			$209,028
	West	E-Commerce	$136,647
			$136,647
4 / E-Home Store	East	E-Commerce	$200,120
			$200,120
	West	E-Commerce	$185,600
			$185,600

Note: Sales are rounded up to the nearest dollar.
Reference: Figure 1.2: Summary Report

2011 Sales - Summary

Category	Sales
Total	**$2,467,105**
Store	
1 / Smith-Fine Goods	$991,322
2 / J.B. Prog & Co.	$744,388
3 / XYZ Warehouse	$345,675
4 / E-Home Store	$385,720
Region	
East	$926,031
West	$756,517
North	$385,438
South	$399,119
Store Type	
E-Commerce	$1,508,362
In-Store	$958,743

Note: Sales are rounded up to the nearest dollar.
Reference: Figure 1.1: Detail Report

Creating Complementary Reports

Objectives

Steps we can implement that will allow for accuracy and
ease of use to the reader:

- A consistent report template
- Consistency of data definitions across reports
- Consistent labels so the end user can easily match items across the reports
- Footnote references clarifying which reports correspond to each other
- Quality checks to ensure that numbers/statistics correspond across reports

Create a Setup Program

- A "setup" program is created to contain the code to be used in both Detail and Summary reports.
- Set up a common ODS Style Template.
- Create Macro Variables to be used in Report Creation.
- Set ODS ESCAPE Character.
- Develop Macro to Modify Incoming Data for Reports.

Set Up a Common ODS Style Template

```
proc template;
  define style styles.journalR;
  parent=styles.journal;

  {Template specifications go here}

end;
run;
```

PROC TEMPLATE Code

```
CLASS FONTS/

  /** Font for Data in Table Cells **/
  "DocFont”=("Georgia", 9 pt)

  /** Font for Column Headers **/
  "HeadingFont”=("Georgia", 16 pt, Bold Italic)

  /** In this example applied to NoteContent **/
  "EmphasisFont”=("Georgia", 9 pt, Bold);
```

PROC TEMPLATE Code Cont…

```
CLASS TABLE/
  rules=none
  borderwidth=2 pt;

CLASS HEADER/
  just=left;

CLASS NOTECONTENT/
/** Applied to COMPUTE block lines and summary rows **/
font=fonts("EmphasisFont");

CLASS SYSTEMTITLE /
  font_size=16 pt
  just=center;

CLASS SYSTEMFOOTER /
  font_size=10 pt
  just=left
```

Create Macro Variables

```
/** TITLES AND FOOTNOTES **/
%let title1    = 2011 Sales;
%let rndfoot   = Note: sales are rounded up to the
nearest dollar.;
%let dreffoot  = Figure 1.2: Summary Report;
%let sreffoot  = Figure 1.1: Detail Report;

/** OPTIONS **/
%let odsopt=nodate nonumber
     orientation=portrait missing=" ";

/** OUTPUT PATH AND ODS STYLE TEMPLATE **/
%let outpath  = c:\clients\project abc\output\;
%let template = journalR;

/** SET ODS ESCAPE CHARACTER **/
ods escapechar = "^";
```

Develop Macro to Modify Incoming Data

```
%macro preproc;
  /** Concatenate Data Sets **/
  data sales(drop=storenam);
    length type $20 storenm $40;
    set store(in=instore)
    ecomm(in=inecomm);

    /** Create Store Type Flags **/
    if inecomm then TYPE = "E-Commerce";
    else if instore then TYPE = "In-Store";

    /** Concatenate Store Number-Name **/
    storenm =
          catx(" / ",put(storenum,3.),storenam);
  run;
%mend preproc;
```

Detail Report

2011 Sales - Detail

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334
	West	E-Commerce	$90,033
		In-Store	$189,356
			$279,389
	North	E-Commerce	$88,024
		In-Store	$99,346
			$187,370
	South	E-Commerce	$110,009
		In-Store	$107,220
			$217,229
2 / J.B. Prog & Co.	East	E-Commerce	$120,050
		In-Store	$89,499
			$209,549

Pre-Processing for Detail Report

```
** Create Template and macro variables;
%include "C:\CLIENTS\Project ABC\Programs\Setup.sas";

** Concatenate and create new data;
%preproc;
```

Partial PROC PRINT of SALES Data

Result of Macro PREPROC

Type	storenm	STORENUM	REG	SALES
In-Store	1 / Smith-Fine Goods	1	East	$200,310
In-Store	1 / Smith-Fine Goods	1	West	$189,356
In-Store	2 / J.B. Prog & Co.	2	East	$89,499
In-Store	2 / J.B. Prog & Co.	2	West	$44,250
In-Store	1 / Smith-Fine Goods	1	North	$99,346
E-Commerce	1 / Smith-Fine Goods	1	East	$107,024
E-Commerce	1 / Smith-Fine Goods	1	West	$90,033
E-Commerce	2 / J.B. Prog & Co.	2	East	$120,050

Pre-Processing for Detail Report

```
** Create Region Informat;
proc format;
  invalue region
    "East"  = 1
    "West"  = 2
    "North" = 3
    "South" = 4;
run;

** Create Region Order Variable;
data saledetr;
   set sales;
   regn=input(reg,region.);
run;
```

SALEDETR Data Snapshot

type	storenm	STORE NUM	REG	SALES	regn
In-Store	1 / Smith-Fine Goods	1	East	$200,310	1
In-Store	1 / Smith-Fine Goods	1	West	$189,356	2
In-Store	2 / J.B. Prog & Co.	2	East	$89,499	1
In-Store	2 / J.B. Prog & Co.	2	West	$44,250	2
E-Commerce	1 / Smith-Fine Goods	1	East	$107,024	1
E-Commerce	1 / Smith-Fine Goods	1	West	$90,033	2

Titles and Footnotes

```
** Set Page Titles and Footnotes;
title1 "&title1 - Detail";
footnote1 "&rndfoot";
footnote2 "Reference: &dreffoot“;
```

Before and After PROC REPORT

```
** ODS RTF Preparation;
ods _all_ close ;
options &odsopt;
ods rtf style=&template file="&outpath.Detail.rtf" bodytitle;

****  {PROC REPORT CODE GOES HERE…} ****;

** Reset ODS Destinations, Titles, Footnotes;
ods rtf close;
ods html;
title;
footnote;
```

PROC REPORT for Detail Report

```
 ** Style Report, Column Data and Headers, and Lines;
proc report data = saledetr nowd center split="|" ;

** Specify Variables to be Reported, in Desired Order;
  column STORENM REGN REG TYPE SALES;

** Define Variable Usage, Labels, and Style Elements;
  define STORENM / order "Store" ;
  define REGN  / noprint order order=internal;
  define REG   / order "Region" ;
  define TYPE  / order "Store Type" ;
  define SALES / "Sales" ;

** Summarize Sales after each Region;
  break after REG / summarize suppress;

** Insert a Blank Line Before each Region;
  compute before REG;
     line " ";
  endcomp;
run;
```

COLUMN Statement Determines Processing Order

- If you want rows ordered first by STORENM, and then by REGN, list the columns in this order.

```
column STORENM REGN REG TYPE SALES;

define STORENM/ order order=formatted "Store”;
define REGN / NOPRINT order order=internal;
define REG  / order order=formatted "Region”;
define TYPE / order  order=formatted"Store Type“;
define SALES/ "Sales”;
```

Overriding PROC TEMPLATE Styles

```
** Apply style overrides in PROC REPORT
Statement;
proc report data = saledetr nowd center
   split="|"
   style(report)=[cellpadding=0 pt]
   style(header)=[vjust=middle
                  borderbottomwidth=2 pt
                  cellheight=.8 in];
```

Example style overrides to specific Columns:

```
define STORENM / order "Store"
  style(column)=[cellwidth=1.4 in
                 font_weight=bold];

define REG     / order "Region"
  style(column)=[cellwidth=1.4 in
                 indent=.55 in]
  style(header)=[indent=.25 in];
```

WITHOUT Style Overrides

2011 Sales - Detail

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334

Column headers wrap

No bottom border

Default Cellpadding

WITH Style Overrides

2011 Sales - Detail

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334

Column headers fit

Bottom border

Cellpadding = 0

Create Summary Lines with Break Statement

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334

```
** Summarize Sales after each Region;
break after REG / summarize suppress;
```

Style Summary Lines

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334

```
** Style summary lines (Sales totals)
break after REG / summarize suppress
   style(summary)=[font_weight=bold
                   foreground=darkblue];
```

The Summary Report

2011 Sales - Summary

Category	Sales
Total	**$2,467,105**
Store	
1 / Smith-Fine Goods	$991,322
2 / J.B. Prog & Co.	$744,388
3 / XYZ Warehouse	$345,675
4 / E-Home Store	$385,720
Region	
East	$926,031
West	$756,517
North	$385,438
South	$399,119
Store Type	
E-Commerce	$1,508,362
In-Store	$958,743

Note: Sales are rounded up to the nearest dollar.
Reference: Figure 1.1: Detail Report

Pre-Processing for Summary Report

```
** Create Template and macro variables;
%include “C:\CLIENTS\Project ABC\ Programs\Setup.sas“ ;

** Concatenate and create new data;
%preproc;
```

Pre-Processing for Summary Report

```
** Create TOTAL variable for Total Row in PROC REPORT;
data salesumr;
  set sales;
  total = "Total";
run ;

** Transpose data to get needed items mapped to
Category;
proc sort data=salesumr;
  by sales storenum;
run;

proc transpose data=salesumr
                    out=salesum2(rename=(col1=catname));
  by sales storenum;
  var total storenm reg type;
run;
```

Transposed Data Set

NAME	STORENUM	catname	SALES	
storenm	1	1 / Smith-Fine Goods	$88,024	
storenm	1	1 / Smith-Fine Goods	$90,033	
storenm	1	1 / Smith-Fine Goods	$99,346	
storenm	1	1 / Smith-Fine Goods	$107,024	
storenm	1	1 / Smith-Fine Goods	$107,220	$991,322 ❶
storenm	1	1 / Smith-Fine Goods	$110,009	
storenm	1	1 / Smith-Fine Goods	$189,356	
storenm	1	1 / Smith-Fine Goods	$200,310	
storenm	2	2 / J.B. Prog & Co.	$44,250	
storenm	2	2 / J.B. Prog & Co.	$79,465	
storenm	2	2 / J.B. Prog & Co.	$89,499	
storenm	2	2 / J.B. Prog & Co.	$97,897	$744,388 ❷
storenm	2	2 / J.B. Prog & Co.	$100,171	
storenm	2	2 / J.B. Prog & Co.	$102,425	
storenm	2	2 / J.B. Prog & Co.	$110,631	
storenm	2	2 / J.B. Prog & Co.	$120,050	

Final Report

2011 Sales - Summary

Category	*Sales*
Total	**$2,467,105**
Store	
1 / Smith-Fine Goods ❶	$991,322
2 / J.B. Prog & Co. ❷	$744,388
3 / XYZ Warehouse	$345,675
4 / E-Home Store	$385,720

Derive CATORD and SUBCTORD for Ordering Rows

catname	catord	subctord
Total	0	0
1 / Smith-Fine Goods	1	1
2 / J.B. Prog & Co.	1	2
3 / XYZ Warehouse	1	3
4 / E-Home Store	1	4
East	2	1002
West	2	1003
North	2	1004
South	2	1005
E-Commerce	3	1007
In-Store	3	1008

Example Code for Creating Order Variables

```
** Informats for Summary Report;
proc format;
  invalue catord
    “TOTAL”          = 0
    “STORENM”     = 1
    “REG”             = 2
    “TYPE”            = 3;

  invalue rowlabl
    "Total"           = 0
    "Region"          = 1
    "East"            = 2
    "West"           = 3
    "North"         = 4
    "South"          = 5
    "Store Type"     = 6
    "E-Commerce” = 7
    "In-Store"       = 8;
run;
```

Example Code for Creating Order Variables

```
data salesum2;
  set salesum2;

  ** Create CATORD for main categories;
  catord = input(upcase(_name_),catord.);

  ** Create SUBCTORD for sub-categories;
  if catord = 0 then subctord = 0;
  else if catord = 1 then subctord = storenum;
  else subctord = input(catname,rowlabl.) + 1000;
run;
```

Data Snapshot Prior to PROC REPORT

catname	catord	subctord	SALES
Total	0	0	$185,600
Total	0	0	$189,356
Total	0	0	$200,120
Total	0	0	$200,310
1 / Smith-Fine Goods	1	1	$88,024
1 / Smith-Fine Goods	1	1	$90,033
1 / Smith-Fine Goods	1	1	$99,346
1 / Smith-Fine Goods	1	1	$107,024
2 / J.B. Prog & Co.	1	2	$44,250
2 / J.B. Prog & Co.	1	2	$79,465
2 / J.B. Prog & Co.	1	2	$89,499
2 / J.B. Prog & Co.	1	2	$97,897

In PROC REPORT, group and order the data

```
COLUMN CATORD SUBCTORD CATNAME SALES;

DEFINE CATORD    / group order=internal noprint;
DEFINE SUBCTORD  / group order=internal noprint;
```

- Order columns and rows by the category number (CATORD) then subcategory number (SUBCTORD)
- GROUP tells PROC REPORT to order and consolidate rows.
- ORDER=INTERNAL tells PROC REPORT to order by unformatted values.

Print Category Row Headers

```
proc format;
   value rowheadr
      0 = " "
      1 = "Store"
      2 = "Region"
      3 = "Store
Type";
run;
```

```
compute before catord;
   line catord rowheadr.;
endcomp;
```

2011 Sales - Summary

Category	Sales
Total	**$2,467,105**
Store	
1 / Smith-Fine Goods	$991,322
2 / J.B. Prog & Co.	$744,388
3 / XYZ Warehouse	$345,675
4 / E-Home Store	$385,720
Region	
East	$926,031
West	$756,517
North	$385,438
South	$399,119
Store Type	
E-Commerce	$1,508,362
In-Store	$958,743

Change Row Style Attributes

Total	**$2,467,105**
Store	
1 / Smith-Fine Goods	$991,322
2 / J.B. Prog & Co.	$744,388
3 / XYZ Warehouse	$345,675
4 / E-Home Store	$385,720

```
compute catname /char length=30;
  if catname ne “Total” then
     do;
       call define(“catname”, ”style”, ”style={indent=.25 in}”);
     end;
  else
    do;
      / ** Total Row **/
      call define(_row_, ”style”, ”style={font_size=10 pt
      font_weight=bold}”);
    end;
endcomp;
```

Recap

- We applied common features to both reports via a Setup program that was called into each individual report program.
- Separate data processing was done for each report to handle specific report needs.
- We used PROC REPORT to apply style overrides, summarize data, order rows, consolidate rows, and create header rows.

Detail and Summary Reports

2011 Sales - Detail

Store	Region	Store Type	Sales
1 / Smith-Fine Goods	East	E-Commerce	$107,024
		In-Store	$200,310
			$307,334
	West	E-Commerce	$90,033
		In-Store	$189,356
			$279,389
	North	E-Commerce	$88,024
		In-Store	$99,346
			$187,370
	South	E-Commerce	$110,009
		In-Store	$107,220
			$217,229
2 / J.B. Prog & Co.	East	E-Commerce	$120,050
		In-Store	$89,499
			$209,549
	West	E-Commerce	$110,631
		In-Store	$44,250
			$154,881
	North	E-Commerce	$97,897
		In-Store	$100,171
			$198,068
	South	E-Commerce	$102,425
		In-Store	$79,465
			$181,890
3 / XYZ Warehouse	East	E-Commerce	$159,902
		In-Store	$49,126
			$209,028
	West	E-Commerce	$136,647
			$136,647
4 / E-Home Store	East	E-Commerce	$200,120
			$200,120
	West	E-Commerce	$185,600
			$185,600

Note: Sales are rounded up to the nearest dollar.
Reference: Figure 1.2: Summary Report

2011 Sales - Summary

Category	Sales
Total	**$2,467,105**
Store	
1 / Smith-Fine Goods	$991,322
2 / J.B. Prog & Co.	$744,388
3 / XYZ Warehouse	$345,675
4 / E-Home Store	$385,720
Region	
East	$926,031
West	$756,517
North	$385,438
South	$399,119
Store Type	
E-Commerce	$1,508,362
In-Store	$958,743

Note: Sales are rounded up to the nearest dollar.
Reference: Figure 1.1: Detail Report

Contact Information

Lisa Fine
United BioSource Corporation
Lisa.Fine@ubc.com